

基于 HT95R5x/6x 的 FSK Decoder 功能的使用方法

文件编码: HA0228S

简介

CID (Calling Identity Delivery), 电话主叫识别讯息发送, 也称为来电显示, 是电信局向被呼叫的电话终端提供的一种的服务。CID 将主叫的电话号码、用户姓名和呼叫时间等信息发给被呼叫端, 被呼叫端接收并储存这些信息, 以使用户查阅。

来电显示实现的一般方法是, 发送端的程控交换机将主叫的电话号码等相关信息通过局间指令系统传递给终端交换机, 终端交换机再将主叫讯息以 FSK (Frequency-shift Keying) 或 DTMF (Dual Tone Multi-Frequency) 方式, 在第一次和第二次振铃间隔期前传递给被呼叫用户终端设备。中国的通信行业, 采用 FSK 方式提供来电显示服务, 当然被呼叫用户之前须申请这项服务。

盛群半导体推出的 CID Phone 8-Bit MCU -- HT95R5x/HT95R6x, 内置 FSK Decoder, 本文将介绍如何使用 HT95R5x/HT95R6x 及其 FSK Decoder 实现来电显示功能。

工作原理

CID 有 Bell 202 FSK 和 ETSI V.23 FSK 等多种规格。其中 Bell 202 是由 Bell 实验室所制订的 FSK CID 规范, 主要在中国、新加坡和美国等国家使用, 本文介绍的是 Bell 202 FSK 的规格。

程控交换机传输 CID 数据的参数如下所示:


调制类型: BFSK (Binary Frequency-Shift Keying)

Mark (Logic 1) : $1200 \pm 12\text{Hz}$

Space (Logic 0) : $2200 \pm 22\text{Hz}$

传输速度 : $1200 \pm 12\text{bps}$

CID 数据的传输可分为两类, 一类是在电话机 On-hook (挂机) 状态下传输, 另一类是在电话机 Off-hook (摘机) 状态下传输。On-hook 状态的 CID 数据传输又可分为两种, 一种是联合振铃的, 另一种是不联合振铃的。下图是 On-hook 状态下联合振铃的 CID 数据传输时序。



FSK Data 包含 Channel Seizure Signal、Mark Signal、Message 和 Checksum。

- Channel Seizure Signal

这是发送 CID 信息时首先要发送的信号，由一组 300 个连续的“0”和“1”交替组成。其第一位为“0”，最后一位为“1”。

- Mark Signal

由 180 个“1”组成。

- Message


Message 的格式有两种，分别是 SDMF (Single Data Message Format) 和 MDMF (Multiple Data Message Format)。程控交换机向客户端传送的信息是 8-bit 的 ASCII 格式的数据。

- Checksum


一个字节的的数据，Message 的所有数据与 Checksum 相加，结果的最后一个字节为 0。

SDMF 和 MDMF 数据格式如下图所示。

SDMF message



MDMF message



下面是一组 FSK SDMF 资料：

04H, 0FH, 30H, 31H, 32H, 33H, 31H, 36H, 35H, 39H, 35H, 36H, 33H, 31H, 39H, 39H, 39H, 39H, D8H。


上述资料中：

- 04H 表示 Message Type, 是 SDMF
- 0FH 表示 Message Length
- 30H、31H、32H、33H 表示日期 01 月 23 号
- 31H、36H、35H、39H 表示时间 16: 59


- 35H、36H、33H、31H、39H、39H、39H、39H 表示电话号码 5631999
- D8H 是 Checksum，用于校验 CID 资料

HT95R5x/HT95R6x 内置 FSK decoder，支持 Bell 202 和 ETSI V.23 两种 CID 的规格，具有 Ring 和 Line Reversal 的侦测功能。FSK Decoder 支持四种能够产生外设中断的中断源，分别是 FSK RAW Data 的下降沿、Ring 或 Line Reversal 侦测、FSK 载波侦测和 FSK 数据封包。

Ring 或 Line Reversal 发生时，内部信号 R_DET 变低，当控制器侦测到 R_DET 的下降沿，FSKS 寄存器的标志位 RDETF 会被设定为 1。当控制器侦测到 FSK 载波信号，标志位 CDETF 被设定为 1。参考线路如下图所示：



HT95R5x/6x 与 FSK 相关的寄存器有 FSKC、FSKS、FSKD、PERIC。




FSKSEL: 选择被封包的 FSK 数据来源，1 为 DOUTC，0 为 DOUT。

CMSK: FSK 载波侦测中断使能，1 为除能，0 为使能。

RMSK: Ring 或 Line Reversal 侦测引起的中断使能，1 为除能，0 为使能。

FMSK: FSK 资料封包引起的中断使能，1 为除能，0 为使能。

F_PWDN: FSK Decoder 电源控制位，1 为断电模式，0 为工作模式。



RINGF: 振铃信号到来的标志位，只读，不能被软件修改。


FSKF: FSK 数据封包中断标志位，1 为 FSK 数据封包动作就绪，0 为未就绪。

DOUTC: FSK Decoder COOK 数据输出，只读，不能被软件修改。

DOUT: FSK RAW 数据输出，只读，不能被软件修改。

CDETF: FSK 载波侦测中断标志位，1 表示侦测到合法的 FSK “0”、“1”信号，0 表示非法的 FSK 信号。

RDETF: Ring 或 Line Reversal 侦测引起的中断标志位，1 表示有侦测到 Ring 或 Line Reversal，0 表示没有侦测到。



FSKD 是放置 FSK 封包的数据。



FSKDF: FSK RAW 数据下降沿引起的中断标志位，中断发生时，设定为 1。

DTMFF: DTMF 接收器中断标志位。

EFSKDI: FSK RAW 资料下降沿引起中断的使能，1 为使能，0 为除能。

EDTMFI: DTMF 接收器中断使能，1 为使能，0 为除能。


需要注意的是，EFSKDI 使能时，控制器会自动将 RMSK、CMSK 和 FMSK 除能。如果中断允许，FSKDF、RDETF、CDETF 和 FSKF 都会引起中断入口地址为 10H 的外设中断，且退出中断服务子程序时，不会被硬件清除为 0，需用软件清除为 0，否则不能进入下一次的中断。

HT95R5x/6x 的 FSK Decoder 模块解析电话线 TIP 和 RING 上的 FSK 信号，并且可产生两种格式的数据，分别是串行数据和 8-bit Packet 数据。为了 FSK Decoder 能够正常工作，需将寄存器 FSKC 中的 F_PWDN 位清除为 0。串行的 FSK 数据用两种格式表示，分别是 RAW Data 和 COOK Data，并且分别可用标志位 DOUT 和 DOUTC 监视。当 Decoder 处于工作模式时，标志位 DOUT 指示 Decoder 的输出，输出的数据流包括 Channel Seizure Signal (由 0 和 1 交替组成)和 Message；标志位 DOUTC 指示 Decoder 的输出与 DOUT 的相似，但不包括 Channel Seizure Signal。若 FSK 信息没有被检测到，则 DOUT 和 DOUTC 都保持高电平。因此用户可用标志位 DOUT 和一个定时器用软件实现的 FSK 译码功能。


除了串行数据的方式外，Decoder 也提供 FSK Packet 数据。当 Decoder 接收到一个 FSK 信号时，Decoder 把 10-bit 的资料打成一个 8-bit 的 packet 数据，10-bit 数据中的第一位和第十位都被忽略。当 8-bit Packet 数据有效时，会被储存到 FSK 数据寄存器 FSKD 中，并且 FSK Packet 数据中断标志位 FSKF 会被设定为 1。如果 FSKC 寄存器的 FMSK 为 0，且外设中断得到允许，这会引起一个外围中断。FSK Decoder 可以将 DOUT 的数据封包，也可将 DOUTC 的数据封包，这由 FSKC 的 FSKSEL 位决定。需要注意的是能够被封包的 10-bit 数据的第一位必须为 0，因而 Mark Signal 的数据不会被封包。

为了能够侦测 FSK 载波信号、解析出串行数据和将 10-bit 数据转成一个 8-bit 的数据包，控制器的必须工作在 Normal 模式。当工作在 Green 和 Sleep 模式时，FSK Decoder 将解出错误的的数据，但 Ring 和 Line Reversal 的侦测功能仍然有效。

FSK Decoder 时序如下图所示。




应用电路图



(参考 HT95R5x data sheet)

S/W 流程图



程序说明

程序设定在 On-hook 状态下接收 SDMF 和 MDMF 格式的 CID 数据，并处理、保存在 RAM Bank1 中。SDMF 的数据包括时间（如 03 月 17 日, 16 时 18 分）和电话号码（如 5551212）；MDMF 的数据包括时间、电话号码和姓名（如 John Smith）。

程序范例

```

; Option: WDT Disable
include HT95R55.inc
ds .section 'data'
VAR0 DB ?
RX DB 40DUP(0) ;save CID information
COUNT0 DB ?
COUNT1 DB ?
COUNT2 DB ?
LENGTH DB ? ; CID Length from CID
LENGTH1 DB ? ; LENGTH1=LENGTH+3,(Compared with LENGTH, Add message type,
; message length and checksum)
FLAG_S DBIT ; Receive Single Data Message Format Flag
FLAG1 DBIT ; Finish receiving flag
FLAG0 DBIT ; Receive channel seizure signal 55H
FLAG_M DBIT ; Receive Multiple Data Message Format Flag
cs .section 'code'

ORG 0000H
JMP MAIN_START
ORG 0004H
RETI
  
```

```

ORG 0008H
RETI
ORG 000CH
RETI
ORG 0010H
JMP FSK_INT
ORG 0014H
RETI
ORG 0018H
RETI

ORG 0020H

MAIN_START:
SET UPEN
CALL DELAY_20MS
SET MODE1
CLR F0
CLR F1 ; System Clock: HCLK 3.58MHz
CALL CLEAR_RAM0
MOV A, 01H
MOV BP, A
CALL CLEAR_RAM1 ; Clear RAM bank0 and bank1
MOV A, OFFSET RX
MOV MP0, A
MOV A, 40H
MOV MP1, A

CLR F_PWDN ; FSK decoder work at operation mode
CLR CMSK ; enable Carrier detect interrupt
CLR RMSK ; enable Ring or line reversal detect interrupt
CLR FMSK ; enable FSK packet data interrupt
CLR EFSKDI ; disable FSK RAW data falling edge interrupt
CLR FSKSEL ; select FSK packet data source DOUT
SET EMI
SET EPERI ; enable peripheral interrupt

L0:
SZ FLAG_S
JMP L1
SZ FLAG_M
JMP L1
JMP L0

L1:
SNZ FLAG1
JMP L0
CLR EPERI
MOV A, 01H
MOV BP, A
MOV A, OFFSET RX
MOV MP0, A

L2:
MOV A, IAR0
MOV IAR1, A ; store the CID from RX to ram bank1
INC MP0
INC MP1
INC COUNT0
MOV A, LENGTH1
SUB A, COUNT0
SNZ C
JMP L3
MOV A, COUNT0

```

```

XOR A, LENGTH1
SNZ Z
JMP L2
INC COUNT2
L3:
CLR FLAG_S
CLR FLAG_M
CLR FLAG1
CLR FLAG0
CLR COUNT0
CLR COUNT1
CLR LENGTH
CLR LENGTH1
MOV A, OFFSET RX
MOV MP0, A
SET EPERI
JMP L0

FSK_INT:
CLR EPERI
SZ CDETF ; check FSK carrier detect interrupt flag
CLR CDETF
SZ RDETF ; check ring or reversal detect interrupt flag
CLR RDETF
SNZ FSKF ; check FSK packet data interrupt flag
JMP EXIT_INT
CLR FSKF
SZ FLAG1
JMP EXIT_INT
CALL DECODE_SUB

EXIT_INT:
CLR PERF
CLR FSKDF
CLR RDETF
CLR CDETF
CLR FSKF
SET EPERI
RETI

DECODE_SUB:
SZ FLAG0
JMP SUB0
MOV A, FSKD
XOR A, 55H
SNZ Z
RET
SET FLAG0 ; detect FSK channel seizure signal 55H

SUB0:
SZ FLAG_S
JMP M1
SZ FLAG_M
JMP M1
MOV A, FSKD
XOR A, 04H ; check SDMF message type
SZ Z
JMP S0
MOV A, FSKD
XOR A, 80H ; check MDMF message type
SZ Z
JMP M0

```

```

RET
S0:
SET FLAG_S
JMP M1

M0:
SET FLAG_M

M1:
INC COUNT1
MOV A, COUNT1
XOR A, 02H
SNZ Z
JMP M2
MOV A, FSKD
MOV LENGTH, A           ; get message length
ADD A, 03H
MOV LENGTH1, A

M2:
MOV A, FSKD
MOV IAR0, A           ; save the CID information to RX
INC MP0
MOV A, COUNT1
SUB A, 3
SNZ C
RET
MOV A, COUNT1
XOR A, LENGTH1
SZ Z
SET FLAG1           ; check if Receive action finish
RET

CLEAR_RAM0:
MOV A, 40H
MOV MP0, A
CLEAR0:
CLR IAR0
MOV A, MP0
XOR A, 0FFH
SZ Z
RET
INC MP0
JMP CLEAR0

CLEAR_RAM1:
MOV A, 40H
MOV MP1, A
CLEAR1:
CLR IAR1
MOV A, MP1
XOR A, 0FFH
SZ Z
RET
INC MP1
JMP CLEAR1

DELAY_20MS:
MOV A, 165
MOV VAR0, A
DELAY:

```

```
SDZ VAR0  
JMP DELAY  
RET
```

结论

本文范例程序可以实现 SDMF 和 MDMF 的 CID 数据译码，用户可参考使用。